Lesson No.26

Keyboard Controller

We will go in further details of the keyboard and its relation to the computer. We will not discuss how the keyboard communicates with the keyboard controller in the computer rather we will discuss how the keyboard controller communicates with the processor. Keyboard is a collection of labeled buttons and every button is designated a number (not the ASCII code). This number is sent to the processor whenever the key is pressed. From this number called the scan code the processor understands which key was pressed. For each key the scan code comes twice, once for the key press and once for the key release. Both are scan codes and differ in one bit only. The lower seven bits contain the key number while the most significant bit is clear in the press code and set in the release code. The IBM PC standard gives a table of the scan codes of all keys. 

If we press Shift-A resulting in a capital A on the screen, the controller has sent the press code of Shift, the press code of A, the release code of A, the release code of Shift and the interrupt handler has understood that this sequence should result in the ASCII code of ‘A’. The ‘A’ key always produces the same scan code whether or not shift is pressed. It is the interrupt handler’s job to remember that the press code of Shift has come and release code has not yet come and therefore to change the meaning of the following key presses. Even the capslock key works the same way.

An interesting thing is that the two shift keys on the left and right side of the keyboard produce different scan codes. The standard way implemented in BIOS is to treat that similarly. That’s why we always think of them as identical. If we leave BIOS and talk directly with the hardware we can differentiate between left and right shift keys with their scan code. Now this scan code is available from the keyboard data port which is 60. The keyboard generates IRQ 1 whenever a key is pressed so if we hook INT 9 and inside it read port 60 we can tell which of the shift keys was hit. Our first program will do precisely this. It will output an L if the left shift key was pressed and R if the right one was pressed. The hooking method is the same as done in the previous chapter. 

	; differentiate left and right shift keys with scancodes

[org 0x0100]

              jmp  start

; keyboard interrupt service routine
kbisr:        push ax

              push es

              mov  ax, 0xb800

              mov  es, ax             ; point es to video memory

              in   al, 0x60           ; read a char from keyboard port
              cmp  al, 0x2a           ; is the key left shift
              jne  nextcmp            ; no, try next comparison
              mov  byte [es:0], 'L'   ; yes, print L at top left
              jmp  nomatch            ; leave interrupt routine
nextcmp:      cmp  al, 0x36           ; is the key right shift
              jne  nomatch            ; no, leave interrupt routine
              mov  byte [es:0], 'R'   ; yes, print R at top left
nomatch:      mov  al, 0x20

              out  0x20, al           ; send EOI to PIC

              pop  es

              pop  ax

              iret

start:        xor  ax, ax

              mov  es, ax             ; point es to IVT base
              cli                     ; disable interrupts
              mov  word [es:9*4], kbisr ; store offset at n*4

              mov  [es:9*4+2], cs     ; store segment at n*4+2
              sti                     ; enable interrupts

l1:           jmp  l1                 ; infinite loop

	033-036
	CLI clears the interrupt flag to disable the interrupt system completely. The processor closes its ears and does not care about the state of the INT pin. Interrupt hooking is done in two instructions, placing the segment and placing the offset. If an interrupt comes inbetween and the vector is ina an indeterminate state, the sytem will go to a junk address and eventually crash. So we stop all interruptions while changing a real time interrupt vector. We set the interrupt flag afterwards to reenable interrupts.

	038
	The program hangs in an infinite loop. The only activity can be caused by a real time interrupt. The kbisr routine is not called from anywhere; it is only automatically invoked as a result of IRQ 1. 


When the program is executed the left and right shift keys can be distinguished with the L or R on the screen. As no action was taken for the rest of the keys, they are effectively disabled and the computer has to be rebooted. To check that the keyboard is actually disabled we change the program and add the INT 16 service 0 at the end to wait for an Esc keypress. As soon as Esc is pressed we want to terminate our program.

	; attempt to terminate program with Esc that hooks keyboard interrupt
[org 0x0100]

              jmp  start

;;;;; COPY LINES 005-029 FROM EXAMPLE 9.2 (kbisr) ;;;;;

start:        xor  ax, ax

              mov  es, ax             ; point es to IVT base
              cli                     ; disable interrupts
              mov  word [es:9*4], kbisr ; store offset at n*4

              mov  [es:9*4+2], cs     ; store segment at n*4+2
              sti                     ; enable interrupts

l1:           mov  ah, 0              ; service 0 – get keystroke
              int  0x16               ; call BIOS keyboard service

              cmp al, 27              ; is the Esc key pressed 
              jne l1                  ; if no, check for next key
              mov ax, 0x4c00          ; terminate program
              int 0x21

	
	


When the program is executed the behavior is same. Esc does not work. This is because the original IRQ 1 handler was written by BIOS that read the scan code, converted into an ASCII code and stored in the keyboard buffer. The BIOS INT 16 read the key from there and gives in AL. When we hooked the keyboard interrupt BIOS is no longer in control, it has no information, it will always see the empty buffer and INT 16 will never return. 

Interrupt Chaining

We can transfer control to the original BIOS ISR in the end of our routine. This way the normal functioning of INT 16 can work as well. We can retrieve the address of the BIOS routine by saving the values in vector 9 before hooking our routine. In the end of our routine we will jump to this address using a special indirect form of the JMP FAR instruction. 

	; another attempt to terminate program with Esc that hooks 

; keyboard interrupt

[org 0x100]

              jmp  start

oldisr:       dd   0                  ; space for saving old isr
; keyboard interrupt service routine
kbisr:        push ax

              push es

              mov  ax, 0xb800

              mov  es, ax             ; point es to video memory

              in   al, 0x60           ; read a char from keyboard port
              cmp  al, 0x2a           ; is the key left shift
              jne  nextcmp            ; no, try next comparison
              mov  byte [es:0], 'L'   ; yes, print L at top left
              jmp  nomatch            ; leave interrupt routine
nextcmp:      cmp  al, 0x36           ; is the key right shift
              jne  nomatch            ; no, leave interrupt routine
              mov  byte [es:0], 'R'   ; yes, print R at top left
nomatch:      ; mov  al, 0x20

              ; out  0x20, al           

              pop es

              pop ax

              jmp far [cs:oldisr]     ; call the original ISR
              ; iret

start:        xor  ax, ax

              mov  es, ax             ; point es to IVT base
              mov ax, [es:9*4]

              mov [oldisr], ax        ; save offset of old routine
              mov ax, [es:9*4+2]

              mov [oldisr+2], ax      ; save segment of old routine
              cli                     ; disable interrupts
              mov  word [es:9*4], kbisr ; store offset at n*4

              mov  [es:9*4+2], cs     ; store segment at n*4+2
              sti                     ; enable interrupts

l1:           mov  ah, 0              ; service 0 – get keystroke
              int  0x16               ; call BIOS keyboard service

              cmp al, 27              ; is the Esc key pressed 
              jne l1                  ; if no, check for next key
              mov ax, 0x4c00          ; terminate program
              int 0x21 

	027-028
	EOI is no longer needed as the original BIOS routine will have it at its end. 

	033
	IRET has been removed and an unconditional jump is introduced. At time of JMP the stack has the exact formation as was when the interrupt came. So the original BIOS routine’s IRET will take control to the interrupted program. We have been careful in restoring every register we modified and retained the stack in the same form as it was at the time of entry into the routine.


When the program is executed L and R are printed as desired and Esc terminates the program as well. Normal commands like DIR work now and shift keys still show L and R as our routine did even after the termination of our program. Now start some application like the editor, it open well but as soon as a key is pressed the computer crashes. 

Actually our hookin and chaining was fine. When Esc was pressed we signaled DOS that our program has terminated. DOS will take all our memory as a result. The routine is still in memory and functioning but the memory is free according to DOS. As soon as we load EDIT the same memory is allocated to EDIT and our routine as overwritten. Now when a key is pressed our routine’s address is in the vector but at that address some new code is placed that is not intended to be an interrupt handler. That may be data or some part of the EDIT program. This results in crashing the computer. 

Unhooking Interrupt

We now add the interrupt restoring part to our program. This code resets the interrupt vector to the value it had before the start of our program. 

	; terminate program with Esc that hooks keyboard interrupt

[org 0x100]

              jmp  start

oldisr:       dd   0                  ; space for saving old isr
;;;;; COPY LINES 005-029 FROM EXAMPLE 9.4 (kbisr) ;;;;;
start:        xor  ax, ax

              mov  es, ax             ; point es to IVT base
              mov ax, [es:9*4]

              mov [oldisr], ax        ; save offset of old routine
              mov ax, [es:9*4+2]

              mov [oldisr+2], ax      ; save segment of old routine
              cli                     ; disable interrupts
              mov  word [es:9*4], kbisr ; store offset at n*4

              mov  [es:9*4+2], cs     ; store segment at n*4+2
              sti                     ; enable interrupts

l1:           mov  ah, 0              ; service 0 – get keystroke
              int  0x16               ; call BIOS keyboard service

              cmp al, 27              ; is the Esc key pressed 
              jne l1                  ; if no, check for next key
              mov ax, [oldisr]        ; read old offset in ax
              mov bx, [oldisr+2]      ; read old segment in bx
              cli                     ; disable interrupts
              mov [es:9*4], ax        ; restore old offset from ax
              mov [es:9*4+2], bx      ; restore old segment from bx
              sti                     ; enable interrupts
              mov ax, 0x4c00          ; terminate program
              int 0x21 


1.1. Terminate and Stay Resident

We change the display to show L only while the left shift is pressed and R only while the right shift is pressed to show the use of the release codes. We also changed that shift keys are not forwarded to BIOS. The effect will be visible with A and Shift-A both producing small ‘a’ but capslock will work. 

There is one major difference from all the programs we have been writing till now. The termination is done using INT 21 service 31 instead of INT 21 service 4C. The effect is that even after termination the program is there and is legally there.

	; TSR to show status of shift keys on top left of screen
[org 0x0100]

              jmp  start

oldisr:       dd   0                  ; space for saving old isr
; keyboard interrupt service routine
kbisr:        push ax

              push es

              mov  ax, 0xb800

              mov  es, ax             ; point es to video memory

              in   al, 0x60           ; read a char from keyboard port
              cmp  al, 0x2a           ; has the left shift pressed
              jne  nextcmp            ; no, try next comparison
              mov  byte [es:0], 'L'   ; yes, print L at first column
              jmp  exit               ; leave interrupt routine
nextcmp:      cmp  al, 0x36           ; has the right shift pressed
              jne  nextcmp2           ; no, try next comparison
              mov  byte [es:0], 'R'   ; yes, print R at second column
              jmp  exit               ; leave interrupt routine
nextcmp2:     cmp  al, 0xaa           ; has the left shift released
              jne  nextcmp3           ; no, try next comparison
              mov  byte [es:0], ' '   ; yes, clear the first column
              jmp  exit               ; leave interrupt routine
nextcmp3:     cmp  al, 0xb6           ; has the right shift released
              jne  nomatch            ; no, chain to old ISR
              mov  byte [es:2], ' '   ; yes, clear the second column
              jmp  exit               ; leave interrupt routine
nomatch:      pop  es

              pop  ax
              jmp  far [cs:oldisr]    ; call the original ISR
exit:         mov  al, 0x20

              out  0x20, al           ; send EOI to PIC
              pop  es

              pop  ax

              iret                    ; return from interrupt
start:        xor  ax, ax

              mov  es, ax             ; point es to IVT base
              mov  ax, [es:9*4]

              mov  [oldisr], ax       ; save offset of old routine
              mov  ax, [es:9*4+2]

              mov  [oldisr+2], ax     ; save segment of old routine
              cli                     ; disable interrupts
              mov  word [es:9*4], kbisr ; store offset at n*4

              mov  [es:9*4+2], cs     ; store segment at n*4+2
              sti                     ; enable interrupts

              mov  dx, start          ; end of resident portion
              add  dx, 15             ; round up to next para

              mov  cl, 4
              shr  dx, cl             ; number of paras 
              mov  ax, 0x3100         ; terminate and stay resident
              int  0x21


When this program is executed the command prompt immediately comes. DIR can be seen. EDIT can run and keypresses do not result in a crash. And with all that left and right shift keys shown L and R on top left of the screen while they are pressed but the shift keys do not work as usual since we did not forwarded the key to BIOS. This is selective chaining.

To understand Terminate and Stay Resident (TSR) programs the DOS memory formation and allocation procedure must be understood. At physical address zero is the interrupt vector table. Above it are the BIOS data area, DOS data area, IO.SYS, MSDOS.SYS and other device drivers. In the end there is COMMAND.COM command interpreter. The remaining space is called the transient program area as programs are loaded and executed in this area and the space reclaimed on their exit. A freemem pointer in DOS points where the free memory begins. When DOS loads a program the freemem pointer is moved to the end of memory, all the available space is allocated to it, and when it exits the freemem pointer comes back to its original place thereby reclaiming all space. This action is initiated by the DOS service 4C. 

The second method to legally terminate a program and give control back to DOS is using the service 31. Control is still taken back but the memory releasing part is modified. A portion of the allocated memory can be retained. So the differene in the two methods is that the freemem pointer goes back to the original place or a designated number of bytes ahead of that old position. Remember that our program crashed because the interrupt routine was overwritten. If we can tell DOS not to reclaim the memory of the interrupt routine, then it will not crash. In the last program we have told DOS to make a number of bytes resident. It becomes a part of the operation system, an extension to it. Just like DOSKEY
 is an extension to the operation system. 

The number of paragraps to reserve is given in the DX register. Paragraph is a unit just like byte, word, and double word. A paragraph is 16 bytes. Therefore we can reserve in multiple of 16 bytes. We write TSRs in such a way that the initialization code and data is located at the end as it is not necessary to make it resident and therefore to save space. 

To calculate the number of paragraphs a label is placed after the last line that is to be made resident. The value of that label is the number of bytes needed to be made resident. A simple division by 16 will not give the correct number of paras as we want our answer to be rounded up and not down. For example 100 bytes should need 7 pages but division gives 6 and a remainder of 4. A standard technique to get rounded up integer division is to add divisor-1 to the dividend and then divide. So we add 15 to the number of bytes and then divide by 16. We use shifting for division as the divisor is a power of 2. We use a form of SHR that places the count in the CL register so that we can shift by 4 in just two instructions instead of 4 if we shift one by one. 

In our program anything after start label is not needed after the program has become a TSR. We can observe that our program has become a part of DOS by giving the following command.

mem /c

This command displays all currently loaded drivers and the current state of memory. We will be able to see our program in the list of DOS drivers.

� DOSKEY is a TSR that shows the previous commands on the command prompt with up and down arrows and allows editing of the command





